

Generalizing Positional Numeral Systems

Luā Tîng-Giān

December 28, 2016

Abstract

Numbers are everywhere in our daily lives, and positional numeral systems are arguably the most important and common representation of numbers. In this work we have constructed a generalized positional numeral system in Agda to model many of these representations, and investigate some of their properties and relationship with the classical unary representation of the natural numbers.

Chapter 1

Introduction

1.1 Positional Numeral Systems

A numeral system is a writing system for expressing numbers, and humans have invented various kinds of numeral systems throughout history. Take the number “2016” for example:

Numeral system	notation
Chinese numerals	兩千零一十六
Roman numerals	MMXVI
	↕↕
	↕↕
	↕
Egyptian numerals	IIIIII

Even so, most of the systems we are using today are positional notations[3] because they can express infinite numbers with just a finite set of symbols called **digits**.

1.1.1 Digits

Any set of symbols can be used as digits as long as we know how to *assign* each digit to the value it represents.

Numeral system	Digits															
decimal	0	1	2	3	4	5	6	7	8	9						
binary	0	1														
hexadecimal	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
Assigned value	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15

We place a bar above a digit to indicate its assignment. For instance, these are the assignments of hexadecimal digits.

$\bar{0} \mapsto 0$	$\bar{1} \mapsto 1$	$\bar{2} \mapsto 2$	$\bar{3} \mapsto 3$
$\bar{4} \mapsto 4$	$\bar{5} \mapsto 5$	$\bar{6} \mapsto 6$	$\bar{7} \mapsto 7$
$\bar{8} \mapsto 8$	$\bar{9} \mapsto 9$	$\bar{A} \mapsto 10$	$\bar{B} \mapsto 11$
$\bar{C} \mapsto 12$	$\bar{D} \mapsto 13$	$\bar{E} \mapsto 14$	$\bar{F} \mapsto 15$

Positional numeral systems represent a number by lining up a series of digits:

$$\xrightarrow{2016}$$

In this case, 6 is called the *least significant digit*, and 2 is known as the *most significant digit*. Except when writing decimal numbers, we will write down numbers in reverse order, from the least significant digit to the most significant digit like this

$$\xleftarrow{6102}$$

1.1.2 Syntax and Semantics

Syntax bears no meaning; its semantics can only be expressed through the process of *converting* to some other syntax. Numeral systems are merely syntax. The same notation can represent different numbers in different context.

Take the notation "11" for example; it could have several meanings.

Numeral system	number in decimal
decimal	11
binary	3
hexadecimal	17

To make things clear, we call a sequence of digits a **numeral**, or **notation**; the number it expresses a **value**, or simply a **number**; the process that converts notations to values an **evaluation**. From now on, **numeral systems** only refer to the positional ones. We will not concern ourselves with other kinds of numeral systems.

1.1.3 Evaluating Numerals

What we mean by a *context* in the previous section is the **base** of a numeral system. The ubiquitous decimal numeral system as we know has the base of 10, while the binaries that can be found in our machines nowadays has the base of 2.

Numeral system	Base	Digits															
decimal	10	0	1	2	3	4	5	6	7	8	9						
binary	2	0	1														
hexadecimal	16	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
Assigned value		0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15

A numeral system of base n has exactly n digits, which are assigned values from 0 to $n - 1$.

Conventionally, the base of a system is annotated by subscripting it to the right of a numeral, like $(2016)_{10}$. We replace the parenthesis with a fancy pair of semantics brackets, like $\llbracket 2016 \rrbracket_{10}$ to emphasize its role as the evaluation function.

To evaluate a notation of a certain base:

$$\llbracket d_0 d_1 d_2 \dots d_n \rrbracket_{base} = \bar{d}_0 \times base^0 + \bar{d}_1 \times base^1 + \bar{d}_2 \times base^2 + \dots + \bar{d}_n \times base^n$$

Where d_n is a digit for all n .

1.2 Unary Numbers and Peano Numbers

Some computer scientists and mathematicians seem to be more comfortable with unary (base-1) numbers because they are isomorphic to the natural numbers à la Peano.

$$\llbracket 1111 \rrbracket_1 \cong \overbrace{\text{suc} (\text{suc} (\text{suc} (\text{suc} + \text{zero})))}^4$$

Statements established on such construction can be proven using mathematical induction. Moreover, people have implemented and proven a great deal of functions and properties on these unary numbers because they are easy to work with.

However, if we are to evaluate unary numerals with the model we have just settled, the only digit of the unary system would have to be assigned to 0 and every numeral would evaluate to zero as a result.

problem The definition of digit assignments can be modified to allow unary digits to start counting from 1, but that would lead to inconsistency among systems of other bases.

Numeral system	Base	Digits															
decimal	10	0	1	2	3	4	5	6	7	8	9						
binary	2	0	1														
hexadecimal	16	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
unary	1		1														
Assigned value		0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15

Moreover, theorems developed on natural numbers à la Peano cannot be migrated to numbers of other representations, although the proofs are mostly identical.

1.3 Binary Numerals in Digital Circuits

Recall how arithmetics such as long addition are performed by hand.

$$\begin{array}{r} 123 \\ + 34 \\ \hline 157 \end{array}$$

The greater a number is, the longer its notation will be, which in terms determines the time it takes to perform operations. Since a system can only have **finitely many** digits, operations such as addition on these digits must be **constant time**. Consequently, the time complexity of operations such as long addition on a numeral would be $O(\lg n)$ at best. The choice of the base is immaterial as long as it is not unary (which would degenerate to $O(n)$).

However, this is not the case for the binary numeral system implemented in arithmetic logic units (ALU). These digital circuits are designed to perform fast arithmetics. Regarding addition, it takes only *constant time*.

It seems that either we have been doing long addition wrong since primary school, or the chip manufacturers have been cheating all the time. But there's a catch! Because we are capable of what is called *arbitrary-precision arithmetic*, i.e., we could perform calculations on numbers of arbitrary size while the binary numbers that reside in machines are bounded by the hardware, which could only perform *fixed-precision arithmetic*.

problem Judging from the time complexity of operations, the binary numerals running in digital circuits is certainly different from the ordinary binary numerals we have known.


1.4 Numerical representation

lists and unary numbers One may notice that the structure of unary numbers looks suspiciously similar to that of lists'. Let's compare their definition in Haskell.

```
data Nat = Zero
         | Suc Nat
```

```
data List a = Nil
            | Cons a (List a)
```

If we replace every `Cons _` with `Suc` and `Nil` with `Zero`, then a list becomes an unary number. This is precisely what the `length` function, a homomorphism from lists to unary numbers, does.




Now let's compare addition on unary numbers and merge (append) on lists:

```
add : Nat → Nat → Nat
add Zero y = y
add (Suc x) y =
    Suc (add x y)
```

```
append : List a → List a → List a
append Nil ys = ys
append (Cons x xs) ys =
    Cons x (append xs ys)
```


Aside from having virtually identical implementations, operations on unary numbers and lists both have the same time complexity. Incrementing a unary number takes $O(1)$, inserting an element into a list also takes $O(1)$; adding two unary numbers takes $O(n)$, appending a list to another also takes $O(n)$.

binomial heaps and binary numbers If we look at implementations and operations of binary numbers and binomial heaps, the resemblances are also uncanny.



The figure above is a binomial heap containing 13 elements.¹ From left to right, there are *binomial trees* of different *ranks* attached to the path that we call “*the spine*”. A binomial heap is composed of binomial trees just as a numeral is composed of digits. If we read the nodes with binomial trees as 1 and those without as 0, then we get the numeral of 13 in binary.

building blocks Single cells in lists and binomial trees in binomial heaps are all different kind of simple data structures that are called *building blocks*. There are also other kinds of building blocks, such as perfect leaf trees.




These building blocks can have different ranks. A binary leaf tree of rank n , for instance, would contain 2^n elements. The data structures we have addressed so far are all composed of a series of building blocks that are ordered by their ranks.

However, these building blocks do not necessarily have to be binary based, as long as multiple building blocks of the same rank can be merged into a building block of a higher rank or vice versa.

¹Nodes that contain elements are painted black.

random access lists and binary numbers Accessing an element on lists typically takes $O(n)$. Instead of using single cells, *random access lists* adopts perfect leaf trees as building blocks. This improves the time complexity of random access from $O(n)$ to $O(\lg n)$ as a random access list would have at most $O(\lg n)$ building blocks, and the tallest perfect leaf tree takes at most $O(\lg \lg n)$ to traverse.




Similar to that of binomial heaps, random access lists also have spines. Also, treating building blocks as digits also yields binary numerals of the container's size.

1.4.1 The correspondence

The strong analogy between data structures and positional numeral systems suggests that numeral systems can serve as templates for designing containers. Such data structures are called **Numerical Representations**[9] [2].

a container of size n	corresponds to	a numeral of n
a building block of rank n	corresponds to	a digit of weight $base^n$
inserting an element	corresponds to	incrementing a numeral
merging two containers	corresponds to	adding two numerals


problem Retrieving the first element (**head**) of a list typically takes only constant time. On the other hand, it takes $O(\lg n)$ on random access lists. To illustrate the problem, consider a random access list with 32 elements:



To access any elements from the list above, we have to skip through four empty nodes before reaching the first building block. Nodes that correspond

to the digit “0” contain no elements. They are not only ineffective but also hinders traversal.

However, if we replace the digits “0” and “1” with “1” and “2”, then the number 32 can be represented as 21111 instead of 000001. By doing so, we eliminate empty nodes and shorten the spine, thus improving the performance of list traversal.



The data structure introduced above is called the *1-2 random access list*. Its presence suggests that a binary numeral system with digits “1” and “2” should be admissible.

Hinze argues[2] that if we add the digit “0” back to the *1-2 random access list*, then the resulting numerical representation, so-called *0-1-2 random access list*, would even have a better performance of insertion and deletion in certain edge cases.

To accommodate these numerical representations, we need a more versatile representation for numeral systems.

1.5 Motivation and Outline

We started off with numerical representations, but then we found that their corresponding numeral systems alone are interesting enough.

- How to capture the numeral systems behind these data structures?
- What are these numeral systems capable of?
- Which kind of numeral systems are suitable for modelling numerical representations? Do they support increment (insertion) or addition (merge)?
- What are the relations between these numeral systems and the natural numbers?

The remainder of the thesis is organized as follows.

Chapter 2 resolves the problems we have addressed in this chapter by proposing some generalizations to the conventional positional numeral systems.

Chapter 3 gives an introduction to *Agda*, the language we use to construct and reason about the representation.

Chapter 4 introduces *equational reasoning* and relevant properties of natural numbers used in the coming chapters.

Chapter ?? constructs the representation for these numeral systems, defines operations such as increment and addition, search for suitable systems for modelling data structures, and examine the relations between these systems and the natural numbers.

Chapter ?? migrates propositions and proofs from the natural numbers à la Peano to our presentation of numbers.

Chapter ?? concludes everything.

Chapter 2

Generalizations

2.1 Base

Recall the evaluation function.

$$\llbracket d_0 d_1 d_2 \dots d_n \rrbracket_{base} = \bar{d}_0 \times base^0 + \bar{d}_1 \times base^1 + \bar{d}_2 \times base^2 + \dots + \bar{d}_n \times base^n$$

Where \bar{d}_n ranges from 0 to $base - 1$ for all n .

As we can see the base of numeral systems has already been generalized. But nonetheless, it is a good start and we will continue to abstract more things away.

2.2 Offset

To cooperate unary numerals, we relax the constraint on the range of digit assignment by introducing a new variable, *offset*:

$$\llbracket d_0 d_1 d_2 \dots d_n \rrbracket_{base} = \bar{d}_0 \times base^0 + \bar{d}_1 \times base^1 + \bar{d}_2 \times base^2 + \dots + \bar{d}_n \times base^n$$

The evaluation of numerals remains the same but the assignment of digits has changed from

$$0, 1, \dots, offset - 1$$

to

$$offset, offset + 1, \dots, offset + base - 1$$

The codomain of the digit assignment function is *shifted* by *offset*. Now that unary numerals would have an offset of 1 and systems of other bases would have offsets of 0.

1-2 binary system Recall *1-2 random access lists* from the previous chapter, which is the numerical representation of a binary numeral system with an offset of 1. Let us see how to count to ten in the 1-2 binary system.¹

Number	Natural	Number	Natural
1	1	6	22
2	2	7	111
3	11	8	211
4	21	9	121
5	12	10	221

There are no restrictions on the symbols of digits. But nonetheless, it is reasonable to choose symbols that match their assigned values, as we choose the symbol “1” and “2” as digits for the 1-2 binary system.

bijective numerations Systems with an offset of 1 are also known as *bijective numerations* because every number can be represented by exactly one numeral. In other words, the evaluation function is bijective. The 1-2 binary system is one such numeration.

zeroless representations A numeral system is said to be *zeroless* if no digits are assigned 0, i.e., offset > 0. Data structures modeled after zeroless systems are called *zeroless representations*. These containers are preferable to their “zeroful” counterparts. Because a digit of value 0 corresponds to a building block with 0 elements, and a building block that contains no element is not only useless, but also hinders traversal as it takes time to skip over these empty nodes, as we have seen in random access lists from the previous chapter.

2.3 Number of Digits

The binary numeral system running in circuits looks different from what we have in hand. Surprisingly, these binary numbers can fit into our representation with just a tweak. If we allow a system to have more digits, then a fixed-precision binary number can be regarded as a single digit! To illustrate

¹As a reminder, the order of digits are reversed.

this, a 32-bit binary number (*Int32*) would become a single digit that ranges from 0 to 2^{32} , while everything else including the base remains the same.

Formerly in our representation, there are exactly *base* number of digits and their assignments range from:

$$\text{offset} \dots \text{offset} + \text{base} - 1$$

By introducing a new index *#digit* to generalize the number of digits, their assignments range from:

$$\text{offset} \dots \text{offset} + \# \text{digit} - 1$$

2.3.1 Redundancy

Numeral systems like *Int32* are said to be **redundant**, because there is more than one way to represent a number. In fact, systems that admit 0 as one of the digits must be redundant, since we can always take a numeral and add leading zeros without changing its value.

Incrementing a decimal numeral such as 999999 takes much more time than 999998 because carries or borrows can cascade. Redundancy provides “buffer” against operations these carries and borrows.

Numerical representations modeled after redundant numeral systems also enjoy similar properties. When designed properly, redundancy can improve the performance of operations of data structures.

2.4 Relations with Natural Numbers

The following table contains all of the numeral systems we have addressed so far, with *base*, *offset*, and *#digit* taken into account. ²

Numeral system	Base	#Digit	Offset
decimal	10	10	0
binary	2	2	0
hexadecimal	16	16	0
unary	1	1	1
1-2 binary	2	2	1
Int32	2	2^{32}	0
Int64	2	2^{64}	0

² *Int32* and *Int64* are respectively 32-bit and 64-bit machine integers.

With great power of generalizations comes poor properties.

Although we are now capable of expressing those numeral systems with just a few indices, there are also some unexpected inhabitant included in this representation.

Chapter 3

A gentle introduction to dependently typed programming in Agda

There are already plenty of tutorials and introductions of Agda [8][7][4]. We will nonetheless compile a simple and self-contained tutorial from the materials cited above, covering the part (and only the part) we need in this thesis.

Some of the more advanced constructions (such as views and universes) will not be introduced in this chapter, but in other places where we need them.

We assume that readers have some basic understanding of Haskell, and those who are familiar with Agda and dependently typed programming may skip this chapter.

3.1 Some basics

Agda is a *dependently typed functional programming language* and also an *interactive proof assistant*. This language can serve both purposes because it is based on *Martin-Löf type theory*[5], hence the Curry-Howard correspondence[10], which states that: "propositions are types" and "proofs are programs." In other words, proving theorems and writing programs are essentially the same. In Agda we are free to interchange between these two interpretations. The current version (Agda2) is a completely rewrite by Ulf Norell during his Ph.D. at Chalmers University of Technology.

We say that Agda is interactive because theorem proving involves a lot of conversations between the programmer and the type checker. Moreover,

it is often difficult, if not impossible, to develop and prove a theorem at one stroke. Just like programming, the process is incremental. So Agda allows us to leave some “holes” in a program, refine them gradually, and complete the proofs “hole by hole”.

Take this half-finished function definition for example.

```
is-zero : ℕ → Bool
is-zero x = ?
```

We can leave out the right-hand side and ask: “what’s the type of the goal?”, “what’s the context of this case?”, etc. Agda would reply us with:

```
GOAL : Bool
x : ℕ
```

Next, we may ask Agda to pattern match on `x` and rewrite the program for us:

```
is-zero : ℕ → Bool
is-zero zero    = ?
is-zero (suc x) = ?
```

We could fulfill these goals by giving an answer, or even ask Agda to solve the problem (by pure guessing) for us if it is not too difficult.

```
is-zero : Int → Bool
is-zero zero    = true
is-zero (suc x) = false
```

After all of the goals have been accomplished and type-checked, we consider the program to be finished. Often, there is not much point in running an Agda program, because it is mostly about compile-time static constructions. This is what programming and proving things looks like in Agda.

3.2 Simply typed programming in Agda

Since Agda was heavily influenced by Haskell, simply typed programming in Agda is similar to that in Haskell.

Datatypes Unlike in other programming languages, there are no “built-in” datatypes such as *Int*, *String*, or *Bool*. The reason is that they can all be created out of thin air, so why bother?

Datatypes are introduced with **data** declarations. Here is a classical example, the type of booleans.

```
data Bool : Set where
  true  : Bool
  false : Bool
```

This declaration brings the name of the datatype (**Bool**) and its constructors (**true** and **false**) into scope. The notation allow us to explicitly specify the types of these newly introduced entities.

1. **Bool** has type **Set**¹
2. **true** has type **Bool**
3. **false** has type **Bool**

Pattern matching Similar to Haskell, datatypes are eliminated by pattern matching. Here is a function that pattern matches on **Bool**.

```
not : Bool → Bool
not true  = false
not false = true
```

Agda is a *total* language, which means that partial functions are not valid constructions. Programmers are obliged to convince Agda that a program terminates and does not crash on all possible inputs. The following example will not be accepted by the termination checker because the case **false** is missing.

```
not : Bool → Bool
not true  = false
```

Inductive datatype Let us move on to a more interesting datatype with an inductive definition. Here is the type of natural numbers.

```
data ℕ : Set where
  zero : ℕ
  suc  : ℕ → ℕ
```

¹**Set** is the type of small types, and **Set**₁ is the type of **Set**, and so on. They form a hierarchy of types.

The decimal number "4" is represented as `suc (suc (suc (suc zero)))`. Agda also accepts decimal literals if the datatype \mathbb{N} complies with certain language pragma.

Addition on \mathbb{N} can be defined as a recursive function.

```
_+_ :  $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$ 
zero + y = y
suc x + y = suc (x + y)
```

We define `_+_` by pattern matching on the first argument, which results in two cases: the base case, and the inductive step. We are allowed to make recursive calls, as long as the type checker is convinced that the function would terminate.

Those underlines surrounding `_+_` act as placeholders for arguments, making it an infix function in this instance.

Dependent functions and type arguments Up till now, everything looks much the same as in Haskell, but a problem arises as we move on to defining something that needs more power of abstraction. Take identity functions for example:

```
id-Bool : Bool → Bool
id-Bool x = x

id- $\mathbb{N}$  :  $\mathbb{N} \rightarrow \mathbb{N}$ 
id- $\mathbb{N}$  x = x
```

In order to define a more general identity function, those concrete types need to be abstracted away. That is, we need *parametric polymorphism*, and this is where dependent types come into play.

A dependent type is a type whose definition may depend on a value. A dependent function is a function whose type may depend on a value of its arguments.

In Agda, function types are denoted as:

```
A → B
```

Where **A** is the type of domain and **B** is the type of codomain. To let **B** depends on the value of **A**, the value has to *named*, in Agda we write:

```
(x : A) → B x
```

The value of **A** is named **x** and then fed to **B**. As a matter of fact, $A \rightarrow B$ is just a syntax sugar for $(_ : A) \rightarrow B$ with the name of the value being irrelevant. The underline **_** here means "I don't bother naming it".

Back to our identity function, if **A** happens to be **Set**, the type of all small types, and the result type happens to be solely **x**:

```
(x : Set) → x
```

Voila, we have polymorphism, and thus the identity function can now be defined as:

```
id : (A : Set) → A → A
id A x = x
```

id now takes an extra argument, the type of the second argument. **id Bool true** evaluates to **true**

Implicit arguments We have implemented an identity function and seen how polymorphism can be modeled with dependent types. However, the additional argument that the identity function takes is rather unnecessary, since its value can always be determined by looking at the type of the second argument.

Fortunately, Agda supports *implicit arguments*, a syntax sugar that could save us the trouble of having to spell them out. Implicit arguments are enclosed in curly brackets in the type expression. We are free to dispense with these arguments when their values are irrelevant to the definition.

```
id : {A : Set} → A → A
id x = x
```

Or when the type checker can figure them out on function application.

```
val : Bool
val = id true
```

Any arguments can be made implicit, but it does not imply that values of implicit arguments can always be inferred or derived from context. We can always make them implicit arguments explicit on application:

```
val : Bool
val = id {Bool} true
```

Or when they are relevant to the definition:

```
silly-not : { _ : Bool } → Bool
silly-not {true} = false
silly-not {false} = true
```

More syntax sugars We could skip arrows between arguments in parentheses or braces:

```
id : {A : Set} (a : A) → A
id {A} x = x
```

Also, there is a shorthand for merging names of arguments of the same type, implicit or not:

```
const : {A B : Set} → A → B → A
const a _ = a
```

Sometimes when the type of some value can be inferred, we could either replace the type with an underscore, say $(A : _)$, or we could write it as $\forall A$. For the implicit counterpart, $\{A : _ \}$ can be written as $\forall \{A\}$.

Parameterized Datatypes Just as functions can be polymorphic, datatypes can be parameterized by other types, too. The datatype of lists is defined as follows:

```
data List (A : Set) : Set where
  [] : List A
  _::_ : A → List A → List A
```

The scope of the parameters spreads over the entire declaration so that they can appear in the constructors. Here are the types of the datatype and its constructors.

```
infixr 5 _::_

[] : {A : Set} → List A
_::_ : {A : Set} → A → List A → List A
List : Set → Set
```

Where A can be anything, even `List (List Bool)`, as long as it is of type `Set`. `infixr` specifies the precedence of the operator `_::_`.

Indexed Datatypes `Vec` is a datatype that is similar to `List`, but more powerful, in that it encodes not only the type of its element but also its length.

```
data Vec (A : Set) : ℕ → Set where
  [] : Vec A zero
  _::_ : {n : ℕ} → A → Vec A n → Vec A (suc n)
```

`Vec A n` is a vector of values of type `A` and has the length of `n`. Here are some of its inhabitants:

```
nil : Vec Bool zero
nil = []

vec : Vec Bool (suc (suc zero))
vec = true :: false :: []
```

We say that `Vec` is *parameterized* by a type of `Set` and is *indexed* by values of \mathbb{N} . We distinguish indices from parameters. However, it is not obvious how they are different by looking at the declaration.

Parameters are *parametric*, in the sense that, they have no effect on the “shape” of a datatype. The choice of parameters only effects which kind of values are placed there. Pattern matching on parameters does not reveal any insights into their whereabouts. Because they are *uniform* across all constructors, one can always replace the value of a parameter with another one of the same type.

On the other hand, indices may affect which inhabitants are allowed in the datatype. Different constructors may have different indices. In that case, pattern matching on indices may yield relevant information about their constructors.

For example, given a term whose type is `Vec Bool zero`, then we are certain that the constructor must be `[]`, and if the type is `Vec Bool (suc n)` for some `n`, then the constructor must be `_::_`.

We could, for instance, define a `head` function that cannot crash.

```
head : ∀ {A n} → Vec A (suc n) → A
head (x :: xs) = x
```

As a side note, parameters can be thought as a degenerate case of indices whose distribution of values is uniform across all constructors.

With abstraction Say, we want to define `filter` on `List`:

```

filter : ∀ {A} → (A → Bool) → List A → List A
filter p [] = []
filter p (x :: xs) = ?

```

We are stuck here because the result of `p x` is only available at runtime. Fortunately, with abstraction allows us to pattern match on the result of an intermediate computation by adding the result as an extra argument on the left-hand side:

```

filter : ∀ {A} → (A → Bool) → List A → List A
filter p [] = []
filter p (x :: xs) with f x
filter p (x :: xs) | true  = x :: filter p xs
filter p (x :: xs) | false = filter p xs

```

Absurd patterns The *unit type*, or *top*, is a datatype inhabited by exactly one value, denoted `tt`.

```

data ⊤ : Set where
  tt : ⊤

```

The *empty type*, or *bottom*, on the other hand, is a datatype that is inhabited by nothing at all.

```

data ⊥ : Set where

```

These types seem useless, and without constructors, it is impossible to construct an instance of `⊥`. What is an type that cannot be constructed good for?

Say, we want to define a safe `head` on `List` that does not crash on any inputs. Naturally, in a language like Haskell, we would come up with a predicate like this to filter out empty lists `[]` before passing them to `head`.

```

non-empty : ∀ {A} → List A → Bool
non-empty []      = false
non-empty (x :: xs) = true

```

The predicate only works at runtime. It is impossible for the type checker to determine whether the input is empty or not at compile time.

However, things are quite different quite in Agda. With *top* and *bottom*, we could do some tricks on the predicate, making it returns a *Set*, rather than a *Bool*!

```

non-empty : ∀ {A} → List A → Set
non-empty []      = ⊥
non-empty (x :: xs) = ⊤

```

Notice that now this predicate is returning a type. So we can use it in the type expression. `head` can thus be defined as:

```

head : ∀ {A} → (xs : List A) → non-empty xs → A
head []      proof = ?
head (x :: xs) proof = x

```

In the $(x :: xs)$ case, the argument `proof` would have type \top , and the right-hand side is simply x ; in the $[]$ case, the argument `proof` would have type \perp , but what should be returned at the right-hand side?

It turns out that, the right-hand side of the $[]$ case would be the least thing to worry about because it is completely impossible to have such a case. Recall that \perp has no inhabitants, so if a case has an argument of that type, it is too good to be true.

Type inhabitation is, in general, an undecidable problem. However, when pattern matching on a type that is obviously empty (such as \perp), Agda allows us to drop the right-hand side and eliminate the argument with `()`.

```

head : ∀ {A} → (xs : List A) → non-empty xs → A
head []      ()
head (x :: xs) proof = x

```

Whenever `head` is applied to some list `xs`, the programmer is obliged to convince Agda that `non-empty xs` reduces to \top , which is only possible when `xs` is not an empty list. On the other hand, applying an empty list to `head` would result in a function of type `head [] : \perp → A` which is impossible to be fulfilled.

Propositions as types, proofs as programs The previous paragraphs are mostly about the *programming* aspect of the language, but there is another aspect to it. Recall the Curry–Howard correspondence, propositions are types and proofs are programs. A proof exists for a proposition the way that a value inhabits a type.

`non-empty xs` is a type, but it can also be thought of as a proposition stating that `xs` is not empty. When `non-empty xs` evaluates to \perp , no value inhabits \perp , which means no proof exists for the proposition \perp ; when `non-empty xs` evaluates to \top , `tt` inhabits \perp , a trivial proof exists for the proposition \top .

In intuitionistic logic, a proposition is considered to be "true" when it is inhabited by a proof, and considered to be "false" when there exists no proof. Contrary to classical logic, where every propositions are assigned one of two truth values. We can see that \top and \perp corresponds to *true* and *false* in this sense.

Negation is defined as a function from a proposition to \perp .

```
¬ : Set → Set
¬ P = P → ⊥
```

We could exploit \perp further to deploy the principle of explosion of intuitionistic logic, which states that: "from falsehood, anything (follows)" (Latin: *ex falso (sequitur) quodlibet*).

```
⊥-elim : ∀ {Whatever : Set} → ⊥ → Whatever
⊥-elim ()
```

Decidable propositions A proposition is decidable when it can be proved or disproved.²

```
data Dec (P : Set) : Set where
  yes : P → Dec P
  no  : ¬ P → Dec P
```

Dec is very similar to its two-valued cousin **Bool**, but way more powerful, because it also explains (with a proof) why a proposition holds or why it does not.

Suppose we want to know if a natural number is even or odd. We know that **zero** is an even number, and if a number is even then its successor's successor is also even.

```
data Even : ℕ → Set where
  base : Even zero
  step : ∀ {n} → Even n → Even (suc (suc n))
```

We also need the opposite of **step** as a lemma.

```
2-steps-back : ∀ {n} → ¬ (Even n) → ¬ (Even (suc (suc n)))
2-steps-back ¬p q = ?
```

²The connective *or* here is not a disjunction in the classical sense. Either way, a proof or a disproval has to be given.

`2-steps-back` takes two arguments instead of one because the return type $\neg (\text{Even } (\text{suc } (\text{suc } n)))$ is actually a synonym of $\text{Even } (\text{suc } (\text{suc } n)) \rightarrow \perp$. Pattern matching on the second argument of type $\text{Even } (\text{suc } (\text{suc } n))$ further reveals that it could only be constructed by `step`. By contradicting $\neg p : \neg (\text{Even } n)$ and $p : \text{Even } n$, we complete the proof of this lemma.

```
contradiction : ∀ {P Whatever : Set} → P → ¬ P → Whatever
contradiction p ¬p = ⊥-elim (¬p p)

two-steps-back : ∀ {n} → ¬ (Even n) → ¬ (Even (suc (suc n)))
two-steps-back ¬p (step p) = contradiction p ¬p
```

Finally, `Even?` determines a number be even by induction on its predecessor's predecessor. `step` and `two-steps-back` can be viewed as functions that transform proofs.

```
Even? : (n : ℕ) → Dec (Even n)
Even? zero          = yes base
Even? (suc zero)    = no (λ ())
Even? (suc (suc n)) with Even? n
Even? (suc (suc n)) | yes p = yes (step p)
Even? (suc (suc n)) | no ¬p = no (two-steps-back ¬p)
```

The syntax of $\lambda ()$ looks weird, as the result of contracting an argument of type \perp of a lambda expression $\lambda x \rightarrow ?$. It is a convention to suffix a decidable function's name with `?`.

Propositional equality Saying that two things are "equal" is a notoriously intricate topic in type theory. There are many different notions of equality [11]. We will not go into each kind of equalities in depth but only skim through those exist in Agda.

Definitional equality, or *intensional equality* is simply a synonym, a relation between linguistic expressions. It is a primitive judgement of the system, stating that two things are the same to the type checker **by definition**.

Computational equality is a slightly more powerful notion. Two programs are consider equal if they compute (beta-reduce) to the same value. For example, `1 + 1` and `2` are equal in Agda in this notion.

However, expressions such as `a + b` and `b + a` are not considered equal by Agda, neither *definitionally* nor *computationally*, because there are simply no rules in Agda saying so.

`a + b` and `b + a` are only *extensionally equal* in the sense that, given **any** pair of numbers, say `1` and `2`, Agda can see that `1 + 2` and `2 + 1` are

computationally equal. But when it comes to **every** pair of numbers, Agda fails to justify that.

We could convince Agda about the fact that $a + b$ and $b + a$ are equal for every pair of a and b by encoding this theorem in a *proposition* and then prove that the proposition holds. This kind of proposition can be expressed with *identity types*.

```
data _≡_ {A : Set} (x : A) : A → Set where
  refl : x ≡ x
```

This inductive datatype says that: for all $a\ b : A$, if a and b are *computationally equal*, that is, both computes to the same value, then `refl` is a proof of $a \equiv b$, and we say that a and b are *propositionally equal*!

`_≡_` is an equivalence relation. It means that `_≡_` is *reflexive* (by definition), *symmetric* and *transitive*.

```
sym : {A : Set} {a b : A} → a ≡ b → b ≡ a
sym refl = refl

trans : {A : Set} {a b c : A} → a ≡ b → b ≡ c → a ≡ c
trans refl refl = refl
```

`_≡_` is congruent, meaning that we could **substitute equals for equals**.

```
cong : {A B : Set} {a b : A} → (f : A → B) → a ≡ b → f a ≡ f b
cong f refl = refl
```

Although these `refl`s look all the same at term level, they are proofs of different propositional equalities.

Dotted patterns Consider an alternative version of `sym` on \mathbb{N} .

```
sym' : (a b : ℕ) → a ≡ b → b ≡ a
sym' a b eq = ?
```

Where `eq` has type $a \equiv b$. If we pattern match on `eq` then Agda would rewrite b as $.a$ and the goal type becomes $a \equiv a$.

```
sym' : (a .a : ℕ) → a ≡ a → a ≡ a
sym' a .a refl = ?
```

What happened under the hood is that a and b are *unified* as the same thing. The second argument is dotted to signify that it is *constrained* by the

first argument **a**. **a** becomes the only argument available for further binding or pattern matching.

Standard library It would be inconvenient if we have to construct everything we need from scratch. Luckily, the community has maintained a standard library that comes with many useful and common constructions.

The standard library is not "chartered" by the compiler or the type checker, there's simply nothing special about it. We may as well as roll our own library.³

³Some primitives that require special treatments, such as IO, are taken care of by language pragmas provided by Agda.

Chapter 4

Properties of Natural Numbers and Equational Reasoning

Properties of natural numbers play a big role in the development of proofs in this thesis. With propositional equality at our disposal, we will demonstrate how to prove properties such as the commutative property of addition. As proofs get more complicated, we will make proving easier by introducing a powerful tool: *equational reasoning*.

4.1 Proving Equational Propositions

Right identity of addition Recap the definition of addition on \mathbb{N} .

```
_+_ : ℕ → ℕ → ℕ
zero + y = y
suc x + y = suc (x + y)
```

`_+_` is defined by induction on the first argument. That means we get the *left identity* of addition for free, as `zero + y` and `y` are *computationally equal*. However, this is not the case for the *right identity*. It has to be proven explicitly.

```
+-right-identity : (n : ℕ) → n + 0 ≡ n
+-right-identity zero    = ?0
+-right-identity (suc n) = ?1
```

By induction on the only argument, we get two sub-goals:

```
?0 : 0 ≡ 0
?1 : suc (n + 0) ≡ suc n
```

`?0` can be trivially proven with `refl`. As for `?1`, we see that its type looks a lot like the proposition we are proving, except that both sides of the equation are “coated” with a `suc`. With `cong suc : ∀ {x y} → x ≡ y → suc x ≡ suc y`, we could substitute a term in `suc` with another if they are equal, and finish the proof by recursively calling itself with a *smaller* argument.

```
+-right-identity : ∀ n → n + 0 ≡ n
+-right-identity zero      = refl
+-right-identity (suc n) = cong suc (+-right-identity n)
```

Moving `suc` to the other side This is an essential lemma for proving more advanced theorems. The proof also follows a similar pattern as that of `+-right-identity`.¹

```
+-suc : ∀ m n → m + suc n ≡ suc (m + n)
+-suc zero      n = refl
+-suc (suc m) n = cong suc (+-suc m n)
```

Commutative property of addition Similarly, by induction on the first argument, we get two sub-goals:

```
+-comm : ∀ m n → m + n ≡ n + m
+-comm zero      n = ?0
+-comm (suc m) n = ?1

?0 : n      ≡ m + zero
?1 : suc (m + n) ≡ m + suc n
```

`?0` can be solved with `+-right-identity` with a “twist”. The symmetry of equality `sym` enables us to swap both sides of an equation.

```
+-comm zero      n = sym (+-right-identity n)
```

However, it is not obvious how to solve `?1` straight out. The proof has to be broken into two steps:

1. Apply `+-suc` with `sym` to the right-hand side of the equation to get `suc (m + n) ≡ suc (n + m)`.

¹In fact, all of these proofs (hence programs) can be generalized with a *fold*, but that is not the point here.

2. Apply the induction hypothesis to `cong suc`.

These small pieces of proofs are glued back together with the transitivity of equality `trans`.

```
+ -comm (suc m) n = trans (cong suc (+ -comm m n)) (sym (+ -suc n m))
```

4.2 Equational Reasoning

We can see that proofs are composable just like programs.

```
trans (cong suc (+ -comm m n)) (sym (+ -suc n m))
```

However, it is difficult to see what is going on in between these clauses, and it could get only worse as propositions get more complicated. Imagine having dozens of `trans`, `sym` and `cong` spreading everywhere.

Fortunately, these complex proofs can be written in a concise and modular manner with a simple yet powerful technique called *equational reasoning*. Agda's flexible mixfix syntax allows the technique to be implemented with just a few combinators[1].

This is best illustrated by an example:

```
+ -comm : ∀ m n → m + n ≡ n + m
+ -comm zero    n = sym (+ -right-identity n)
+ -comm (suc m) n =
  begin
    suc m + n
  ≡( refl )
    suc (m + n)
  ≡( cong suc (+ -comm m n) )
    suc (n + m)
  ≡( sym (+ -suc n m) )
    n + suc m
  ■
```

With equational reasoning, we can see how an expression equates with another, step by step, justified with theorems. The first and the last step corresponds to two sides of the equation of a proposition. `begin_` marks the beginning of a reasoning; `_≡(_)_` chains two expressions with the justification placed in between; `_■` marks the end of a reasoning (*QED*).

4.2.1 Anatomy of Equational Reasoning

A typical equational reasoning can often be broken down into **three** parts.

1. Starting from the left-hand side of the equation, through a series of steps, the expression will be “arranged” into a form that allows the induction hypothesis to be applied. In the following example of `+-comm`, nothing needs to be arranged because these two expressions are computationally equal (the `refl` can be omitted).

```
begin
  suc m + n
≡( refl )
  suc (m + n)
```

2. `m + n` emerged as part of the proposition which enables us to apply the induction hypothesis.

```
      suc (m + n)
≡( cong suc (+-comm m n) )
      suc (n + m)
```

3. After applying the induction hypothesis, the expression is then “rearranged” into the right-hand side of the equation, hence completes the proof.

```
      suc (n + m)
≡( sym (+-suc n m) )
      n + suc m
■
```

arranging expressions To arrange an expression into the shape that we desire as in part 1 and part 3, while remaining equal. We need properties such as commutativity or associativity of some operator, or distributive properties when there is more than one operator.

The operators we will be dealing with often comes with these properties. Take addition and multiplication, for example, together they form a nice semiring structure.

substituting equals for equals As what we have seen in 2, sometimes there is only a part of an expression needs to be substituted. Say, we have a proof `eq : X ≡ Y`, and we want to substitute `X` for `Y` in a more complex expression `a b (c X) d`. We could ask `cong` to "target" the part to substitute by supplying a function like this:

```
λ w → a b (c w) d
```

Which abstracts the part we want to substitute away, such that:

```
cong (λ w → a b (c w) d) eq : a b (c X) d ≡ a b (c Y) d
```

4.3 Decidable Equality on Natural Numbers

Equality is decidable on natural numbers, which means that we can always tell whether two numbers are equal, and explain the reason with a proof.

```
_≐_ : Decidable {A = ℕ} _≡_
zero ≐ zero    = yes refl
suc m ≐ suc n   with m ≐ n
suc m ≐ suc .m | yes refl = yes refl
suc m ≐ suc n   | no prf   =
  no (prf ∘ (λ p → subst (λ x → m ≡ pred x) p refl))
zero ≐ suc n    = no λ()
suc m ≐ zero    = no λ()
```

Decidable functions are often used together with *with-abstractions*.

```
answer : ℕ → Bool
answer n with n ≐ 42
answer n | yes p = true
answer n | no ¬p = false
```

Where `p : n ≡ 42` and `¬p : n ≠ 42`.

4.4 Preorder

Aside from stating that two expressions are equal, a proposition can also state that one expression is "less than or equal to" than another, given a preorder.

A preorder is a binary relation that is *reflexive* and *transitive*. Often denoted as `_≤_`, such a binary relation on natural numbers is defined as:

```

data _≤_ : ℕ → ℕ → Set where
  z≤n : ∀ {n} → zero ≤ n
  s≤s : ∀ {m n} (m≤n : m ≤ n) → suc m ≤ suc n

```

The following is a proof of $3 \leq 5$:

```

3≤5 : 3 ≤ 5
3≤5 = s≤s (s≤s (s≤s z≤n))

```

To prove $3 \leq 5$, we need a proof of $2 \leq 4$ for $s \leq s$, and so on, until it reaches zero where it ends with a $z \leq n$.

Here are some other binary relations than can be defined with $_ \leq _$.

```

_<_ : Rel ℕ Level.zero
m < n = suc m ≤ n

```

```

_≠_ : Rel ℕ Level.zero
a ≠ b = ¬ a ≤ b

```

```

_≥_ : Rel ℕ Level.zero
m ≥ n = n ≤ m

```

4.5 Preorder reasoning

Combinators for equational reasoning can be further generalized to support *preorder reasoning*. Preorders are *reflexive* and *transitive*, which means that expressions can be chained with a series of relations just as that of equational reasoning.

Suppose we already have $m \leq m+n : \forall m n \rightarrow m \leq m + n$ and we want to prove a slightly different theorem.

```

m≤n+m : ∀ m n → m ≤ n + m
m≤n+m m n =
  start
    m
  ≤( m≤m+n m n )
    m + n
  ≈( +-comm m n )
    n + m
□

```

Where $_ \leq (_) _$ and $_ \approx (_) _$ are respectively transitive and reflexive combinators.² Step by step, starting from the left-hand side of the relation, expressions get greater and greater as it reaches the right-hand side the relation.

monotonicity of operators In equational reasoning, we could substitute part of an expression with something equal with **cong** because $_ \equiv _$ is congruent. However, we cannot substitute part of an expression with something *greater* in general.

Take the following function **f** as example.

```
f : ℕ → ℕ
f 0 = 1
f 1 = 0
f _ = 1
```

f returns 1 on all inputs except for 1. $0 \leq 1$ holds, but it does not imply that $f\ 0 \leq f\ 1$ also holds. As a result, a generic mechanism like **cong** does not exist in preorder reasoning. We can only substitute part of an expression when the function is *monotonic*.

4.6 Decidable Preorder on Natural Numbers

Preorder is also decidable on natural numbers, which means that we can always tell whether one number is less than or equal to another.

```
_ ≤? _ : Decidable _ ≤_
zero ≤? _      = yes z ≤ n
suc m ≤? zero  = no λ()
suc m ≤? suc n with m ≤? n
...           | yes m ≤ n = yes (s ≤ s m ≤ n)
...           | no  m > n = no  (m > n ◦ ≤-pred)
```

With with-abstractions we could define some function like this:

```
threshold : ℕ → ℕ
threshold n with n ≤? 87
threshold n | yes p = n
threshold n | no  ¬p = 87
```

Where $p : n \leq 42$ and $\neg p : n \not\leq 42$.

²Combinators for preorder reasoning are renamed to prevent conflictions with equational reasoning.

4.7 Skipping trivial proofs

From now on, we will dispense with most of the steps and justifications in equational and preorder reasonings, because it is often obvious to see what happened in the process.

In fact, there are no formal distinction between the proofs we disregard and those we feel important. They are all equally indispensable to Agda.

4.8 Relevant Properties of Natural Numbers

Relevant properties of \mathbb{N} used in the remainder of the thesis are introduced in this section.

Aside from some basic properties taken from the standard library, we have also added some useful theorems, lemmata, and corollaries.³

4.8.1 Equational Propositions

natural number

```
data  $\mathbb{N}$  : Set where
  zero :  $\mathbb{N}$ 
  suc  :  $\mathbb{N} \rightarrow \mathbb{N}$ 
```

- `cancel-suc` : $\forall \{x\ y\} \rightarrow \text{suc } x \equiv \text{suc } y \rightarrow x \equiv y$
suc is injective.

addition

```
 $\_ + \_$  :  $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$ 
zero  + y = y
suc x + y = suc (x + y)
```

- `+-right-identity` : $\forall n \rightarrow n + 0 \equiv n$
the right identity of addition.

³*Theorem, lemma, corollary* and *property* are all synonyms for *established proposition*. There are no formal distinction between these terms and they are used exchangeably in the thesis.

- $\text{+-suc} : \forall m n \rightarrow m + \text{suc } n \equiv \text{suc } (m + n)$
moving `suc` from one term to another.
- $\text{+-assoc} : \forall m n o \rightarrow (m + n) + o \equiv m + (n + o)$
the associative property of addition.
- $\text{+-comm} : \forall m n \rightarrow m + n \equiv n + m$
the commutative property of addition.
- $\text{[a+b]+c}\equiv\text{[a+c]+b} : \forall a b c \rightarrow a + b + c \equiv a + c + b$
a convenient corollary for swapping terms.
- $\text{a+[b+c]}\equiv\text{b+[a+c]} : \forall a b c \rightarrow a + (b + c) \equiv b + (a + c)$
a convenient corollary for swapping terms.
- $\text{cancel-+-left} : \forall i \{j k\} \rightarrow i + j \equiv i + k \rightarrow j \equiv k$
the left cancellation property of addition.
- $\text{cancel-+-right} : \forall k \{i j\} \rightarrow i + k \equiv j + k \rightarrow i \equiv j$
the right cancellation property of addition.

multiplication

```

_ * _ : ℕ → ℕ → ℕ
zero * y = y
suc x * y = y + (x * y)

```

- $\text{* -right-zero} : \forall n \rightarrow n * 0 \equiv 0$
the right absorbing element of multiplication.
- $\text{* -left-identity} : \forall n \rightarrow 1 * n \equiv n$
the left identity of addition multiplication.
- $\text{* -right-identity} : \forall n \rightarrow n * 1 \equiv n$
the right identity of addition multiplication.
- $\text{+-*-suc} : \forall m n \rightarrow m * \text{suc } n \equiv m + m * n$
multiplication over `suc`.
- $\text{* -assoc} : \forall m n o \rightarrow (m * n) * o \equiv m * (n * o)$
the associative property of multiplication.
- $\text{* -comm} : \forall m n \rightarrow m * n \equiv n * m$
the commutative property of multiplication.

- `distribr-*-+` : $\forall m\ n\ o \rightarrow (n + o) * m \equiv n * m + o * m$
the right distributive property of multiplication over addition.
- `distrib-left-*-+` : $\forall m\ n\ o \rightarrow m * (n + o) \equiv m * n + m * o$
the left distributive property of multiplication over addition.

monus

Monus, or *truncated subtraction*, is a kind of subtraction that never goes negative when the subtrahend is greater than the minued.

```
_÷_ : Nat → Nat → Nat
n ÷ zero = n
zero ÷ suc m = zero
suc n ÷ suc m = n ÷ m
```

- $0 \div n \equiv 0$: $\forall n \rightarrow 0 \div n \equiv 0$
- $n \div n \equiv 0$: $\forall n \rightarrow n \div n \equiv 0$
- $m + n \div n \equiv m$: $\forall m\ n \rightarrow (m + n) \div n \equiv m$
- $m + n \div m \equiv n$: $\forall \{m\ n\} \rightarrow m \leq n \rightarrow m + (n \div m) \equiv n$
- $m \div n + n \equiv m$: $\forall \{m\ n\} \rightarrow n \leq m \rightarrow m \div n + n \equiv m$
- `÷-+-assoc` : $\forall m\ n\ o \rightarrow (m \div n) \div o \equiv m \div (n + o)$
the associative property of monus and addition.
- `+--÷-assoc` : $\forall m\ \{n\ o\} \rightarrow o \leq n \rightarrow (m + n) \div o \equiv m + (n \div o)$
the associative property of monus and addition.
- `*-distrib-÷r` : $\forall m\ n\ o \rightarrow (n \div o) * m \equiv n * m \div o * m$
the right distributive property of monus over multiplication.

min and max

So called `min` and `max` in Haskell. `Min _Π_` computes the lesser of two numbers.

```
_Π_ : ℕ → ℕ → ℕ
zero Π n      = zero
suc m Π zero  = zero
suc m Π suc n = suc (m Π n)
```

Max \sqcup computes the greater of two numbers.

```

 $\sqcup$  :  $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$ 
zero  $\sqcup$  n      = n
suc m  $\sqcup$  zero = suc m
suc m  $\sqcup$  suc n = suc (m  $\sqcup$  n)

```

- \sqcap -comm : $\forall m n \rightarrow m \sqcap n \equiv n \sqcap m$
the commutative property of min.
- \sqcup -comm : $\forall m n \rightarrow m \sqcup n \equiv n \sqcup m$
the commutative property of max.

4.8.2 Relational Propositions

natural number

- \leq -pred : $\forall \{m n\} \rightarrow \text{suc } m \leq \text{suc } n \rightarrow m \leq n$
inverse of $s \leq s$.
- $\leq \Rightarrow \leq$: $_ \leq _ \Rightarrow _ \leq _$
- $\leq \Rightarrow \neq$: $_ \leq _ \Rightarrow _ \neq _$
- $\leq \Rightarrow \star$: $_ \leq _ \Rightarrow _ \star _$
- $\leq \Rightarrow \neq$: $_ \leq _ \Rightarrow _ \neq _$
- $\leq \Rightarrow \neq$: $_ \leq _ \Rightarrow _ \neq _$
- $\leq \Rightarrow \neq$: $_ \leq _ \Rightarrow _ \neq _$
- $\leq \wedge \neq \Rightarrow <$: $\forall \{m n\} \rightarrow m \leq n \rightarrow m \neq n \rightarrow m < n$
- $\geq \wedge \neq \Rightarrow >$: $\forall \{m n\} \rightarrow m \geq n \rightarrow m \neq n \rightarrow m > n$

addition

- \leq -step : $\forall \{m n\} \rightarrow m \leq n \rightarrow m \leq 1 + n$
- \leq -steps : $\forall \{m n\} k \rightarrow m \leq n \rightarrow m \leq k + n$
- $m \leq m+n$: $\forall m n \rightarrow m \leq m + n$

- $n \leq m+n : \forall m\ n \rightarrow n \leq m + n$
- $_+-mono_ : \forall \{m_1\ m_2\ n_1\ n_2\} \rightarrow m_1 \leq m_2 \rightarrow n_1 \leq n_2 \rightarrow m_1 + n_1 \leq m_2 + n_2$
the monotonicity of addition
- $n+-mono : \forall \{i\ j\}\ n \rightarrow i \leq j \rightarrow n + i \leq n + j$
 $_+-mono_$ with the first argument fixed.
- $+n-mono : \forall \{i\ j\}\ n \rightarrow i \leq j \rightarrow n + i \leq n + j$
 $_+-mono_$ with the second argument fixed.
- $n+-mono-inverse : \forall n \rightarrow \forall \{a\ b\} \rightarrow n + a \leq n + b \rightarrow a \leq b$
the inverse of $n+-mono$
- $+n-mono-inverse : \forall n \rightarrow \forall \{a\ b\} \rightarrow a + n \leq b + n \rightarrow a \leq b$
the inverse of $+n-mono$
- $+mono-contr : \forall \{a\ b\ c\ d\} \rightarrow a \geq b \rightarrow a + c < b + d \rightarrow c < d$

multiplication

monus

min and max

Chapter 5

Constructions [still writing]

The representation for positional numeral systems in Agda will be constructed in this chapter, along with the generalizations introduced in section 2.

- **base**: the base of a numeral system, denoted **b**.
- **#digit**: the number of digits, denoted **d**.
- **offset**: the number where the digits starts from, denoted **o**.

5.1 Digit: the basic building block

As the basic building block of numerals, we will demonstrate how to choose a suitable representation for digits in this section.

5.1.1 Fin

To represent a digit, we use a datatype that is conventionally called *Fin* which can be indexed to have some exact number of inhabitants.

```
data Fin : ℕ → Set where
  zero : {n : ℕ} → Fin (suc n)
  suc  : {n : ℕ} (i : Fin n) → Fin (suc n)
```

The definition of **Fin** looks the same as **ℕ** on the term level, but different on the type level. The index of a **Fin** increases with every **suc**, and there can only be at most **n** of them before reaching **Fin (suc n)**. In other words, **Fin n** would have exactly *n* inhabitants.

5.1.2 Definition

Digit is simply just a synonym for **Fin**, indexed by the number of digits **d** of a system. Since the same digit may represent different values in different numeral systems, it is essential to make the context clear.

```
Digit : ℕ → Set
Digit d = Fin d
```


Ordinary binary digits for example can thus be represented as:

```
Binary : Set
Binary = Digit 2

零 : Binary
零 = zero

一 : Binary
一 = suc zero
```

5.1.3 Digit Assignment



Digits are assigned to \mathbb{N} together with the offset o of a system, ranging from o to $d + o$.

```
Digit-toℕ : ∀ {d} → Digit d → ℕ → ℕ
Digit-toℕ x o = toℕ x + o
```

¹

However, not all natural numbers can be converted to digits. The value has to be in a certain range, between o and $d + o$. Values less than o are increased to o . Values greater than $d + o$ are prohibited by the supplied upper-bound.

```
Digit-fromℕ : ∀ {d}
  → (n o : ℕ)
  → (upper-bound : d + o ≥ n)
```

¹ **toℕ** : $\forall \{n\} \rightarrow \text{Fin } n \rightarrow \mathbb{N}$
converts from **Fin** n to \mathbb{N} .

```

    → Digit (suc d)
Digit-fromℕ {d} n o upper-bound with n ÷ o ≤? d
Digit-fromℕ {d} n o upper-bound | yes p = fromℕ≤ (s≤s p)
Digit-fromℕ {d} n o upper-bound | no ¬p = contradiction p ¬p
  where   p : n ÷ o ≤ d
         p = start
              n ÷ o
         ≤( +n-mono o upper-bound )
         (d + o) ÷ o
         ≈( m÷n÷n≡m d o )
         d
    □

```

2

properties

$$\mathbb{N} \begin{array}{c} \xrightarrow{\text{Digit-from}\mathbb{N}} \\ \xleftarrow{\text{Digit-to}\mathbb{N}} \end{array} \text{Digit } d$$

Digit-fromℕ-toℕ states that the value of a natural number should remain the same, after converting back and forth between **Digit** and \mathbb{N} .

```

Digit-fromℕ-toℕ : ∀ {d o}
  → (n : ℕ)
  → (lower-bound : o ≤ n)
  → (upper-bound : d + o ≥ n)
  → Digit-toℕ (Digit-fromℕ {d} n o upper-bound) o ≡ n
Digit-fromℕ-toℕ {d} {o} n lb ub with n ÷ o ≤? d
Digit-fromℕ-toℕ {d} {o} n lb ub | yes q =
  begin
    toℕ (fromℕ≤ (s≤s q)) + o
  ≡( cong (λ x → x + o) (toℕ-fromℕ≤ (s≤s q)) )
    n ÷ o + o
  ≡( m÷n÷n≡m lb )
    n
  ■
Digit-fromℕ-toℕ {d} {o} n lb ub | no ¬q = contradiction q ¬q
  where   q : n ÷ o ≤ d
         q = +n-mono-inverse o (
              start
                n ÷ o + o
              ≈( m÷n÷n≡m lb )
            )

```

² **fromℕ≤** : $\forall \{m n\} \rightarrow m < n \rightarrow \text{Fin } n$
 converts from \mathbb{N} to **Fin** n given the number is small enough.

$$\leq(\text{ub } n \text{ } d + o)$$

$$\square)$$

3

Digits have a upper-bound and a lower-bound after evaluation.

```
Digit-upper-bound : ∀ {d} → (o : ℕ) → (x : Digit d) → Digit-toℕ x o < d + o
Digit-upper-bound {d} o x = +n-mono o (bounded x)

Digit-lower-bound : ∀ {d} → (o : ℕ) → (x : Digit d) → Digit-toℕ x o ≥ o
Digit-lower-bound {d} o x = m≤n+m o (toℕ x)
```

4

5.1.4 Operations


Increment

To increment a digit, the digit must not be *the greatest*.

```
digit+1 : ∀ {d}
  → (x : Digit d)
  → (¬greatest : ¬ (Greatest x))
  → Fin d
digit+1 x ¬greatest =
  fromℕ≤ {suc (toℕ x)} (≤Λ≠⇒< (bounded x) ¬greatest)
```

Where $\leq\Lambda\neq\Rightarrow< (\text{bounded } x) \neg\text{greatest} : \text{suc } (\text{to}\mathbb{N} \ x) < d$.

properties



³ $\text{to}\mathbb{N}\text{-from}\mathbb{N}\leq : \forall \{m\ n\} \ (m < n : m < n) \rightarrow \text{to}\mathbb{N} \ (\text{from}\mathbb{N}\leq m < n) \equiv m$
states that a number should remain the same after converting back and forth.

⁴ $\text{bounded} : \forall \{n\} \ (i : \text{Fin } n) \rightarrow \text{to}\mathbb{N} \ i < n$
a property about the upper-bound of a $\text{Fin } n$.

A digit taking these two routes should result in the same \mathbb{N} .

```

digit+1-to $\mathbb{N}$  :  $\forall \{d\} o$ 
  → (x : Digit d)
  → (¬greatest : ¬ (Greatest x))
  → Digit-to $\mathbb{N}$  (digit+1 x ¬greatest) o  $\equiv$  suc (Digit-to $\mathbb{N}$  x o)
digit+1-to $\mathbb{N}$  {d} {o} x ¬greatest =
  begin
    Digit-to $\mathbb{N}$  (digit+1 x ¬greatest) o
 $\equiv$  ( cong (λ w → w + o) (to $\mathbb{N}$ -from $\mathbb{N} \leq (\leq \wedge \# \Rightarrow < (\text{bounded } x) \neg\text{greatest})) )$ 
    suc (Digit-to $\mathbb{N}$  x o)
  ■

```

Increase then Subtract


Increases a digit and then subtract it by n . This function is useful for implementing *carrying*. When the digit to increase is already the greatest, we have to subtract it by an amount (usually the base) after the increment.

```

digit+1-n :  $\forall \{d\}$ 
  → (x : Digit d)
  → Greatest x
  → (n :  $\mathbb{N}$ )
  → n > 0
  → Digit d
digit+1-n x greatest n n>0 =
  from $\mathbb{N} \leq$  (digit+1-n-lemma x greatest n n>0)

```

properties



A digit taking these two routes should result in the same \mathbb{N} .

```

digit+1-n-to $\mathbb{N}$  :  $\forall \{d\} o$ 
  → (x : Digit d)
  → (greatest : Greatest x)
  → (n :  $\mathbb{N}$ )
  → (n>0 : n > 0)
  → n  $\leq$  d
  → Digit-to $\mathbb{N}$  (digit+1-n x greatest n n>0) o  $\equiv$  suc (Digit-to $\mathbb{N}$  x o) - n

```

```

digit+1-n-toℕ {zero} {o} () greatest n n>0 n≤d
digit+1-n-toℕ {suc d} {o} x greatest n n>0 n≤d =
  begin
    toℕ (digit+1-n x greatest n n>0) + o
  ≡( cong (λ w → w + o) (toℕ-fromℕ≤ (digit+1-n-lemma x greatest n n>0)) )
    suc (toℕ x) ÷ n + o
  ≡( +-comm (suc (toℕ x) ÷ n) o )
    o + (suc (toℕ x) ÷ n)
  ≡( sym (++-assoc o {suc (toℕ x)} {n} (
    start
      n
    ≤( n≤d )
      suc d
    ≈( sym greatest )
      suc (toℕ x)
    □)
  )
    (o + suc (toℕ x)) ÷ n
  ≡( cong (λ w → w ÷ n) (++-comm o (suc (toℕ x))) )
    suc (toℕ x) + o ÷ n
  ■)

```

5.1.5 Special Digits

The Greatest Digit

constructions The greatest digit of a system is constructed by converting the index d to Fin .

```

greatest-digit : ∀ d → Digit (suc d)
greatest-digit d = fromℕ d

```

5

predicates Judging whether a digit is the greatest by converting it to \mathbb{N} . This predicate also comes with a decidable version.

```

Greatest : ∀ {d} (x : Digit d) → Set
Greatest {d} x = suc (toℕ x) ≡ d

Greatest? : ∀ {d} (x : Digit d) → Dec (Greatest x)
Greatest? {d} x = suc (toℕ x) ≐ d

```

⁵ $\text{from}\mathbb{N} : \forall \{n\} \rightarrow \text{Fin } n \rightarrow \mathbb{N}$

construct the greatest possible $\text{Fin } n$ when given an index n .

properties Converting from the greatest digit to \mathbb{N} should result in $d + o$.

```
greatest-digit-to $\mathbb{N}$  :  $\forall \{d\ o\}$ 
   $\rightarrow (x : \text{Digit } (\text{suc } d))$ 
   $\rightarrow \text{Greatest } x$ 
   $\rightarrow \text{Digit-to}\mathbb{N} \ x \ o \equiv d + o$ 
greatest-digit-to $\mathbb{N}$   $\{d\} \{o\} \ x \ \text{greatest} = \text{cancel-suc } ($ 
   $\text{begin}$ 
     $\text{suc } (\text{Digit-to}\mathbb{N} \ x \ o)$ 
   $\equiv ( \text{refl } )$ 
     $\text{suc } (\text{to}\mathbb{N} \ x) + o$ 
   $\equiv ( \text{cong } (\lambda w \rightarrow w + o) \ \text{greatest} )$ 
     $\text{suc } d + o$ 
   $\blacksquare )$ 
```

A digit is the greatest if and only if it is greater than or equal to all other digits. This proposition is proven by induction on both of the compared digits.

```
greatest-of-all :  $\forall \{d\} \ (o : \mathbb{N}) \rightarrow (x \ y : \text{Digit } d)$ 
   $\rightarrow \text{Greatest } x$ 
   $\rightarrow \text{Digit-to}\mathbb{N} \ x \ o \geq \text{Digit-to}\mathbb{N} \ y \ o$ 
greatest-of-all  $o \ \text{zero} \ \text{zero} \ \text{refl} = \leq\text{-refl}$ 
greatest-of-all  $o \ \text{zero} \ (\text{suc } ()) \ \text{refl}$ 
greatest-of-all  $o \ (\text{suc } x) \ \text{zero} \ \text{greatest}$ 
   $= +n\text{-mono } o \ \{\text{zero}\} \ \{\text{suc } (\text{to}\mathbb{N} \ x)\} \ z \leq n$ 
greatest-of-all  $o \ (\text{suc } x) \ (\text{suc } y) \ \text{greatest}$ 
   $= s \leq s \ (\text{greatest-of-all } o \ x \ y \ (\text{cancel-suc } \text{greatest}))$ 
```

The Carry

A carry is a digit that is transferred to a more significant digit to compensate the “loss” of the original digit.

constructions The carry is defined as the greater of these two values:

- the least digit of a system
- the digit that is assigned to 1

In case that the least digit is assigned to 0, rendering the carry useless. Since the least digit is determined by the offset o , the value of the carry is defined as follows.

```

carry : ℕ → ℕ
carry o = 1 ∪ o

```

And then we construct the carry by converting `carry o` to `Digit`:

```

carry-digit : ∀ d o → 2 ≤ suc d + o → Digit (suc d)
carry-digit d o proper =
  Digit-fromℕ
    (carry o)
    0
    (carry-upper-bound {d} proper)

```

properties The value of the carry should remain the same after converting back and forth.

```

carry-digit-toℕ : ∀ d o
  → (proper : 2 ≤ suc (d + o))
  → Digit-toℕ (carry-digit d o proper) o ≡ carry o
carry-digit-toℕ d o proper
  = Digit-fromℕ-toℕ
    (carry o)
    (m≤nℓm o 1)
    (carry-upper-bound {d} proper)

```

The carry also have an upper-bound and a lower-bound, similar to that of `Digit`.

```

carry-lower-bound : ∀ {o} → carry o ≥ o
carry-lower-bound {o} = m≤nℓm o 1

carry-upper-bound : ∀ {d o} → 2 ≤ suc d + o → carry o ≤ d + o
carry-upper-bound {d} {zero} proper = ≤-pred proper
carry-upper-bound {d} {suc o} proper = n≤m+n d (suc o)

```

5.2 Num: a representation for positional numeral systems

5.2.1 Definition

Numerals in positional numeral systems are composed of sequences of digits. So the definition of `Numeral` will be similar to that of `List`, except that a

`Numeral` must contain at least one digit while a list may contain no elements at all. The most significant digit is placed in `_•` while the least significant digit is placed at the end of the sequence. `Numeral` is indexed by all three generalizations.

```
infixr 5 _::_

data Numeral : ℕ → ℕ → ℕ → Set where
  _•   : ∀ {b d o} → Digit d → Numeral b d o
  _::_ : ∀ {b d o} → Digit d → Numeral b d o → Numeral b d o
```

The decimal number “2016” for example can be represented as:

```
MMXVI : Numeral 10 10 0
MMXVI = # 6 :: # 1 :: # 0 :: (# 2) •
```

Where `#_ : ∀ m {n} {m < n : True (suc m ≤? n)} → Fin n` converts from `ℕ` to `Fin n` given the number is small enough.

To extract the least significant digit from a numeral:

```
lsd : ∀ {b d o} → (xs : Numeral b d o) → Digit d
lsd (x •   ) = x
lsd (x :: xs) = x
```

5.2.2 Converting to natural numbers

Converting to natural numbers is fairly trivial.

```
[_] : ∀ {b d o} → (xs : Numeral b d o) → ℕ
[ ] { } { } {o} (x • ) = Digit-toℕ x o
[ ] {b} { } {o} (x :: xs) = Digit-toℕ x o + [ xs ] * b
```

5.3 Dissecting Numeral Systems with Views

There are many kinds of numeral systems inhabit in `Numeral`. These systems have different interesting properties and they should be treated differently, so we sort them into **four categories** accordingly.

Systems with no digits at all The number of digits of a system is determined by the index `d`. If `d` happens to be 0, then there will be no digits in any of these systems. Although they seem useless, these systems have plenty

of properties. Since there are not digits at all, any property that is related to digits would hold vacuously.

Systems with the base of 0 If b , the base of a system, happens to be 0, then only the least significant digit would have effects on the evaluation, because the rest of the digits would diminish into nothing.

```
[[ x •      ]] = Digit-toℕ x o
[[ x :: xs ]] = Digit-toℕ x o + [[ xs ]] * 0
```

Systems with only zeros Consider when d is set to 1 and o set to 0. There will be one digit but the only digit is assigned to 0.

```
0, 00, 000, 0000, ...
```

As a result, every numeral would evaluate to 0 regardless of the base.

“Proper” systems The rest of systems that does not fall into any of the categories above are considered *proper*.

5.3.1 Categorizing Systems

These “categories” are represented with a datatype called **NumView** that is indexed by the three indices: b , d , and o .

```
data NumView : (b d o : ℕ) → Set where
  NullBase    : ∀ d o → NumView 0      (suc d) o
  NoDigits    : ∀ b o → NumView b      0      o
  AllZeros    : ∀ b → NumView (suc b) 1      0
  Proper      : ∀ b d o → (proper : suc d + o ≥ 2)
                        → NumView (suc b) (suc d) o
```

By pattern matching on indices, different configurations of indices are sorted into different **NumViews**.

```
numView : ∀ b d o → NumView b d o
numView b      zero      o      = NoDigits b o
numView zero   (suc d)    o      = NullBase d o
numView (suc b) (suc zero) zero  = AllZeros b
numView (suc b) (suc zero) (suc o) = Proper b zero (suc o) _
numView (suc b) (suc (suc d)) o    = Proper b (suc d) o _
```

Together with *with-abstractions*, we can, for example, define a function to determine whether a numeral system is boring or not:

```
boring : ∀ b d o → Bool
boring b d o with numView b d o
boring _ _ _ | NullBase d o          = false
boring _ _ _ | NoDigits b o          = false
boring _ _ _ | AllZeros b            = false
boring _ _ _ | Proper b d o proper = true
```

As we can see, the function `numView` does more than sorting indices into different categories. It also reveals relevant information and properties about these categories. For instance, if a system `Numeral b d o` is classified as *Proper*, then we know that:

- `b` is greater than 0.
- `d` is also greater than 0.
- `o` can be any value as long as $d + o \geq 2$; we name this requirement *proper*.

5.3.2 Views

The sole purpose of `NumView` is to sort out and expose some interesting properties about its indices. Such datatypes are called *views*[12] as they present different aspects of the very same object. Functions like `numView` are called *view functions* or *eliminators*[6] because they provide different ways of eliminating a datatype.

Views are **reusable** as they free us from having to pattern match on the same indices or data again and again. On the other hand, they can be customized to our needs since they are just *ordinary functions*. We will define more views and use them extensively in the coming sections.

5.4 Special Properties of Categories

NoDigits Although systems with no digits have no practical use, they are pretty easy to deal with because all properties related to digits would hold unconditionally for systems of `NoDigits`. This is proven by deploying *the principle of explosion*.

```
NoDigits-explode : ∀ {b o a} {Whatever : Set a}
```

```

→ (xs : Numeral b 0 o)
→ Whatever
NoDigits-explode (() •   )
NoDigits-explode (() :: xs)

```

NullBase The theorem below states that, evaluating a numeral of **NullBase** and evaluating its least significant digit would have the same result.

```

toN-NullBase : ∀ {d o}
→ (x : Digit d)
→ (xs : Numeral 0 d o)
→ [ x :: xs ] ≡ Digit-toN x o
toN-NullBase {d} {o} x xs =
begin
  Digit-toN x o + [ xs ] * 0
≡⟨ cong (λ w → Digit-toN x o + w) (*-right-zero [ xs ]) ⟩
  Digit-toN x o + 0
≡⟨ +-right-identity (Digit-toN x o) ⟩
  Digit-toN x o
■

```

AllZeros The theorem below states every numeral of **AllZeros** would evaluate to 0 regardless of the base. To exploit the fact that there is only one digit in such numerals, we pattern match on the digit to eliminate other possible cases.

```

toN-AllZeros : ∀ {b} → (xs : Numeral b 1 0) → [ xs ] ≡ 0
toN-AllZeros (z •   ) = refl
toN-AllZeros (s () •   )
toN-AllZeros {b} (z :: xs)
= cong (λ w → w * b) (toN-AllZeros xs)
toN-AllZeros (s () :: xs)

```

5.5 The Maximum Numeral

A number is said to be the *maximum* if there are no other numbers greater than itself.

```

Maximum : ∀ {b d o} → (xs : Numeral b d o) → Set
Maximum {b} {d} {o} xs = (ys : Numeral b d o) → [ xs ] ≥ [ ys ]


```

If a numeral is the maximum ⁶ then its least significant digit (LSD) must be the greatest.

```
Maximum⇒Greatest-LSD : ∀ {b} {d} {o}
  → (xs : Numeral b d o)
  → Maximum xs
  → Greatest (lsd xs)
```

5.5.1 Properties of each Category

NullBase




It is obvious that systems of **NullBase** have maxima. If a numeral's LSD happens to be the greatest, then the numeral must be the maximum.

```
Maximum-NullBase-Greatest : ∀ {d} {o}
  → (xs : Numeral 0 (suc d) o)
  → Greatest (lsd xs)
  → Maximum xs
```

With this lemma, we can tell whether a numeral is the maximum by looking at its LSD. In case that the LSD is not the greatest, we could say no by contraposition.

```
Maximum-NullBase : ∀ {d} {o}
  → (xs : Numeral 0 (suc d) o)
  → Dec (Maximum xs)
Maximum-NullBase xs with Greatest? (lsd xs)
Maximum-NullBase xs | yes greatest =
  yes (Maximum-NullBase-Greatest xs greatest)
Maximum-NullBase | no ¬greatest =
  no (contraposition (Maximum⇒Greatest-LSD xs) ¬greatest)
```

AllZeros



⁶More precisely, “If a value of a numeral is the maximum ...”

All numerals of systems of **AllZeros** are maxima since they are all mapped to 0.

```
Maximum-AllZeros : ∀ {b}
  → (xs : Numeral b 1 0)
  → Maximum xs
Maximum-AllZeros xs ys = reflexive (
  begin
    [ ys ]
  ≡( toN-AllZeros ys )
    zero
  ≡( sym (toN-AllZeros xs) )
    [ xs ]
  ■)
```

Proper On the contrary, there are no maxima in the systems of **Proper**. In fact, that is the reason why they are categorized as *proper* in the first place. This theorem is proven by contradicting two propositions:

- Given **claim** : **Maximum** **xs**, we claim that **xs** is greater than or equal to **greatest-digit d :: xs**, a numeral we composed by prefixing it with the greatest digit.
- On the other hand, We prove that **xs** is less than **greatest-digit d :: xs**.

```
Maximum-Proper : ∀ {b d o}
  → (xs : Numeral (suc b) (suc d) o)
  → (proper : 2 ≤ suc (d + o))
  → ¬ (Maximum xs)
Maximum-Proper {b} {d} {o} xs proper claim = contradiction p ¬p
  where
    p : [ xs ] ≥ [ greatest-digit d :: xs ]
    p = claim (greatest-digit d :: xs)
    ¬p : [ xs ] < [ greatest-digit d :: xs ]
    ¬p = <=> (
      start
        suc [ xs ]
      ≈( cong suc (sym (*-right-identity [ xs ])) )
        suc ([ xs ] * 1)
      ≤( s≤s (n*-mono [ xs ] (s≤s z≤n)) )
        suc ([ xs ] * suc b)
      ≤( +n-mono ([ xs ] * suc b) (≤-pred proper) )
        d + o + [ xs ] * suc b
      ≈( cong
        (λ w → w + [ xs ] * suc b)
```

```

      (sym (greatest-digit-to $\mathbb{N}$  (Fin.from $\mathbb{N}$  d)
      (greatest-digit-is-the-Greatest d)))
    )
  [[ greatest-digit d :: xs ]]
□)

```

5.5.2 Determine the Maximum

We can *decide* whether a numeral is the maximum by applying them to lemmata of each categories.

```

Maximum? :  $\forall$  {b d o}
  → (xs : Numeral b d o)
  → Dec (Maximum xs)
Maximum? {b} {d} {o} xs with numView b d o
Maximum? xs | NullBase d o = Maximum-NullBase xs
Maximum? xs | NoDigits b o = no (NoDigits-explode xs)
Maximum? xs | AllZeros b   = yes (Maximum-AllZeros xs)
Maximum? xs | Proper b d o proper = no (Maximum-Proper xs proper)

```

5.6 Systems with an Upper Bound

5.7 The Next Numeral

5.7.1 Incrementing a Numeral

5.7.2 Continuous Systems

Bibliography

- [1] P. V. Erik Hesselink. Equational reasoning in agda. Presentation slides, sep 2008.
- [2] R. Hinze et al. Numerical representations as higher order nested datatypes. Technical report, Citeseer, 1998.
- [3] D. E. Knuth. The art of computer programming. volume 2, seminumerical algorithms. 1998.
- [4] J. Malakhovski. Brutal [meta]introduction to dependent types in agda, mar 2013.
- [5] P. Martin-Lef. Intuitionistic type theory. *Naples: Bibliopolis*, 76, 1984.
- [6] C. McBride and J. McKinna. The view from the left. *J. Funct. Program.*, 14(1):69–111, Jan. 2004.
- [7] S.-C. Mu. Dependently typed programming. Lecture handouts, jul 2016.
- [8] U. Norell. Dependently typed programming in agda. In *Advanced Functional Programming*, pages 230–266. Springer, 2009.
- [9] C. Okasaki. *Purely functional data structures*. PhD thesis, Citeseer, 1996.
- [10] M. H. Sørensen and P. Urzyczyn. *Lectures on the Curry-Howard isomorphism*, volume 149. Elsevier, 2006.
- [11] T. B. Urs Schreiber. equality, dec 2016.
- [12] P. Wadler. Views: A way for pattern matching to cohabit with data abstraction. In *Proceedings of the 14th ACM SIGACT-SIGPLAN symposium on Principles of programming languages*, pages 307–313. ACM, 1987.